
Abstract Starting with an evaluator for a language, an abstract ma-
chine for the same language can be mechanically derived using successive
program transformations. This has relevance to studying both the time
and space properties of programs because these can be estimated by
counting transitions of the abstract machine and measuring the size of
the additional data structures needed, such as environments and stacks.
In this paper we will use this process to derive a function that accurately
counts the number of steps required to evaluate expressions in a simple
language, and illustrate this function with a range of examples.

1 Introduction

The problem of reasoning about intensional properties of functional programs,
such as the time requirements, is a long-running one. It is complicated by dif-
ferent evaluation strategies and sharing of expressions, meaning that some parts
of a program may not be run or only partially so. One of the issues involved in
reasoning about the amount of time a program will take to complete is what to
count as an atomic unit in evaluation, or an evaluation step.

An evaluator is usually an implementation of the denotational semantics
[13] of a language — it evaluates an expression based on the meaning of its
sub-expressions. This level of understanding helps us reason about extensional
properties of the language, but it doesn’t say anything about the underlying
way that the evaluation is taking place. By contrast, an operational semantics
[10] shows us the method that is being used to evaluate an expression, and the
conventional approach is to use this to measure the number of steps that are
required. This, however, may not be very accurate because what is usually being
measured is β-reductions, each of which may take arbitrarily long. An example
of this approach is using the tick monad [15], which counts β-reductions.

It is proposed that a more realistic measure would be to count transitions
in an actual machine. The idea is to get more detailed information than by
just counting β-reductions, but still have a principled way of obtaining this
information. This would not be as accurate as actual time measurements, but
would provide a useful half-way house between these two approaches. What is
needed is a justification that the machine correctly implements the evaluation
semantics and a way to reflect the number of steps required by the machine back
to the semantic level.



Recently Danvy et al have explored the basis of abstract machines, and the
process of deriving them from evaluators [1, 2, 3, 5]. This process uses, in par-
ticular, two program transformation techniques: transformation to continuation
passing style, to make order of evaluation explicit; and defunctionalization, to
eliminate the consequent use of higher-order functions.

In this paper, we will apply this process to the problem of accurately counting
evaluation steps. In brief, this consists of introducing a simple language in which
to write expressions, an evaluator for this language, and then deriving a cor-
responding abstract machine using successive program transformations. Simple
step counting is then added to the machine, by threading and incrementing a
counter that measures the number of transitions.

The next stage is to apply the same process but in the reverse order, res-
ulting in an evaluator that additionally counts the number of steps, directly
corresponding to the number of transitions of the underlying abstract machine.
Finally, a direct step counting function will be calculated from this evaluator and
will be used to reason about evaluation of some example computations expressed
in the language.

A particular aspect of our derivation process is that the program at each stage
is calculated directly from a specification of its behaviour [8]. All the programs
are written in Haskell [9], but no specific features of this language are used, so
they may be easily adapted to other functional languages.

2 A Simple Language

To start with we will consider a simple language with expressions consisting of
integers and addition, and with integers as values:

data Expr = Add Expr Expr | Val Int
type Value = Int

Although this language is not powerful enough to be used to analyse the time re-
quirements of any interesting computations, it will be sufficient to show the deriv-
ation process without over-complication. An extended language will be presented
later to look at some example functions.

2.1 Evaluator

The initial evaluator takes an expression and evaluates it to a value:

eval :: Expr → Value
eval (Val v) = v
eval (Add x y) = eval x + eval y

That is, evaluating an integer value returns that integer, and evaluating an ad-
dition evaluates both sides of the addition to an integer and then adds them
together. The order of evaluation is not specified at this level but will be de-
termined by the semantics of the underlying language; in particular, when the
expression is an addition, Add x y , the evaluation of the x or y may occur first.



2.2 Tail-recursive Evaluator

Our aim is to turn the evaluator into an abstract machine, a term-rewriting
system that makes explicit the step-by-step process by which evaluation can be
performed. More precisely, we seek to construct an abstract machine implemen-
ted in Haskell as a first-order, tail-recursive function.

The evaluator is already first order, but it is not tail-recursive. It can be made
so by transforming it to continuation passing style (CPS) [11]. A continuation is
a function that represents the rest of a computation; this makes the evaluation
order of the arguments explicit, so intermediate results need to be ordered using
the continuation. A program can be transformed in to CPS by redefining it to
take an extra argument, a function which is applied to the result of the original
one. In our case, the continuation function will take an argument of type Value
and its result is a Value:

type Con = Value → Value

The new tail-recursive evaluator can be calculated from the old one by using the
specification:

evalTail :: Expr → Con → Value
evalTail e c = c (eval e)

That is, the new evaluator has the same behaviour as simply applying the con-
tinuation to the result of the original evaluator. The definition of this function
can be calculated by performing induction on the structure of the expression, e.

Case : e = Val v

evalTail (Val v) c
= { specification }

c (eval v)
= { definition of eval }

c v

Case : e = Add x y

evalTail (Add x y) c
= { specification }

c (eval (Add x y))
= { definition of eval }

c (eval x + eval y)
= { reverse β-reduction, abstract over eval x }

(λm → c (m + eval y)) (eval x )
= { inductive assumption for x }

evalTail x (λm → c (m + eval y))
= { reverse β-reduction, abstract over eval y }

evalTail x (λm → (λn → c (m + n)) (eval y))
= { inductive assumption for y }

evalTail x (λm → evalTail y (λn → c (m + n)))



In conclusion, we have calculated the following recursive definition:

evalTail :: Expr → Con → Value
evalTail (Val v) c = c v
evalTail (Add x y) c = evalTail x (λm →

evalTail y (λn → c (m + n)))

In the case when the expression is an integer the continuation is simply applied
to the integer value. In the addition case, the first argument to the addition is
evaluated first, with the result being passed in to a continuation. The second
expression argument is then evaluated inside the continuation, with its result
being passed in to an inner continuation. Both integer results are added together
in the body of this function and the original continuation is applied to the result.

The evaluator is now tail recursive, in that the right hand side is a direct
recursive call and there is nothing to be done after the call returns. In making
the evaluator tail-recursive we have introduced an explicit evaluation order: the
evaluation of the addition now has to occur in left-to-right order.

The semantics of the original evaluation function can be recovered by sub-
stituting in the identity function for the continuation:

eval e = evalTail e (λv → v)

2.3 Abstract Machine

The next step is to make the evaluator first order. This is done by defunctionaliz-
ing the continuations [11]. At the moment, the continuations are functions of the
type Value → Value, but the whole function space is not required: the continu-
ation functions are only created in three different ways. Defunctionalization is
performed by looking at all places where functions are made and replacing them
with a new data structure that takes as arguments any free variables required.

The data structure required is as follows,

data Cont = Top for the initial continuation (λv → v)
| AddL Cont Expr for (λm → evalTail y (...))
| AddR Value Cont for (λn → c (m + n))

The reason for the constructor names is that the data structure is the structure
of evaluation contexts for the language [6]. It could alternatively be viewed as a
stack, pushing expressions still to be evaluated and values to be saved.

To recover the functionality of the continuation we define an apply function
which has the same semantics for each instance of the continuation function:

apply :: Cont → Con
apply Top = λv → v
apply (AddR m c) = λn → apply c (m + n)
apply (AddL c y) = λm → evalTail y (apply (AddL c m))

We now seek to construct a new evaluator, evalMach, that behaves in the same
way as evalTail , except that it uses representations of continuations, rather than



real continuations; that is, we require evalMach e c = evalTail e (apply c). From
this specification, the definition of evalMach can be calculated by induction on e:

Case : e = Val v

evalMach (Val v) c
= { specification }

evalTail (Val v) (apply c)
= { definition of evalTail }

apply c v

Case : e = Add x y

evalMach (Add x y) c
= { specification }

evalTail (Add x y) (apply c)
= { definition of evalTail }

evalTail x (λm → evalTail y (λn → apply c (m + n)))
= { definition of apply }

evalTail x (λm → evalTail y (apply (AddR m c)))
= { definition of apply }

evalTail x (apply (AddL c y))
= { inductive assumption, for x }

evalMach x (AddL c y)

We have now calculated the following recursive function:

evalMach :: Expr → Cont → Value
evalMach (Val v) c = apply c v
evalMach (Add x y) c = evalMach x (AddL c y)

Evaluating an integer calls the apply function with the current context and the
integer value. Evaluating an addition evaluates the first argument and stores the
second with the current context using the AddL constructor.

Moving the λ-abstracted terms to the left and applying the specification in
the AddL case, gives the following revised definition for apply :

apply :: Cont → Value → Value
apply Top v = v
apply (AddR m c) n = apply c (m + n)
apply (AddL c y) m = evalMach y (AddR m c)

The apply function takes a context and a value and returns the value if the
context is Top. When the context is AddR this represents the case when both
sides of the addition have been evaluated, so the results are added together
and the current context is applied to the result. The AddL context represents
evaluating the second argument to the addition, so the evalMach function is
called and the result from the first argument and the current context saved
using the AddR context.



The original semantics can be recovered by passing in the equivalent of the
initial continuation, the Top constructor:

eval e = evalMach e Top

2.4 Step Counting Machine

The number of time steps required to evaluate an expression is to be measured
by counting the number of transitions of the abstract machine. The abstract ma-
chine derived can be simply modified by adding a step count that is incremented
each time a transition, a function call to evalMach or apply , is made. The step
count is added as an accumulator, rather than just incrementing the count that
the recursive call returns, so that it is still an abstract machine.

It is relevant here to note that this is just one possible derivation to produce
an abstract machine. Different abstract machines may be generated by applying
different program transformations, as demonstrated in [5].

type Step = Int
stepMach :: (Expr ,Step)→ Cont → (Value,Step)
stepMach (Val v , s) c = apply ′ c (v , s + 1)
stepMach (Add x y , s) c = stepMach (x , s + 1) (AddL c y)
apply ′ :: Cont → (Value,Step)→ (Value,Step)
apply ′ Top (v , s) = (v , s + 1)
apply ′ (AddL c y) (m, s) = stepMach (y , s + 1) (AddR m c)
apply ′ (AddR m c) (n, s) = apply ′ c (m + n, s + 1)

In this case we are only counting the machine transitions and the actual
addition of the integers is defined to happen instantly, though this could be
extended by introducing an additional factor that represents the number of steps
to perform an addition.

The evaluation function now returns a pair, where the first part is the eval-
uated value, and the second is the number of steps taken, which is initialised
to zero. Therefore, the semantics of the original evaluator can be recovered by
taking the first part of the pair returned:

eval e = fst (stepMach e 0 Top)

2.5 Step Counting Tail-recursive Evaluator

The aim now is to derive a function that counts the number of steps required to
evaluate an expression. The specification for this is the second part of the pair
returned by the abstract machine:

steps e = snd (stepMach (e, 0) Top)

The first stage in the reverse process is to refunctionalize the representation of the
continuation. The original continuation was a function of type Value → Value,
so the type of the new one will be (Value,Step)→ (Value,Step).



type Con ′ = (Value,Step)→ (Value,Step)

Again, this can be calculated by induction on e, from the following specification:

stepTail (e, s) (apply ′ c′) = evalMach (e, s) c′

The refunctionalized version is:

stepTail :: (Expr ,Step)→ Con ′ → (Value,Step)
stepTail (Val v) s c = c (v , s + 1)
stepTail (Add x y) s c = stepTail x (s + 1) (λ(m, s ′)→

stepTail y (s ′ + 1) (λ(n, s ′′)→ c (m + n, s ′′ + 1)))

The step counting semantics can be redefined as:

steps e = snd (stepTail (e, 0) ((v , s)→ (v , s + 1)))

Now, the same program transformations are performed in the reverse order to
derive an evaluator that counts steps at the evaluator level, corresponding to the
number of transitions of the abstract machine.

2.6 Step Counting Evaluator with Accumulator

The step counting evaluator can be transformed from CPS back to direct style
by calculation, using the following specification to remove the continuation:

c (stepAcc (e, s)) = stepTail (e, s) c

The resulting evaluator is:

stepAcc :: (Expr ,Step)→ (Value,Step)
stepAcc (Val v) s = (v , s + 1)
stepAcc (Add x y) s = let (m, s ′) = stepAcc (x , s + 1)

(n, s ′′) = stepAcc (y , s ′ + 1)
in (m + n, s ′′ + 1)

The new step counting function becomes:

steps e = snd (stepAcc (e, 0)) + 1

2.7 Step Counting Evaluator

At the moment the step counting evaluator treats the step count as an accumu-
lator. This can be removed, by calculating a new function without one, using
the specification:



stepEval e = let (v , s ′) = stepAcc (e, s)
in (v , s ′ − s)

Again, this can be calculated by induction over the structure of the expression,
to give the new step counting evaluator:

stepEval :: Expr → (Value,Step)
stepEval (Val v) = (v , 1)
stepEval (Add x y) = let (m, s) = stepEval x

(n, s ′) = stepEval y
in (m + n, s + s ′ + 3)

The semantics of the steps function can be expressed as:

steps e = snd (stepEval e) + 1

2.8 Step Counting Function

The final stage is to calculate a standalone steps function. This will take an
expression and return the number of steps to evaluate the expression, calling
the original evaluator when the result of evaluation is required. The resulting
function can be produced by routine calculation:

steps e = steps ′ e + 1
steps ′ (Val v) = 1
steps ′ (Add x y) = steps ′ x + steps ′ y + 3

The derived steps function shows that that evaluation of an expression is an
auxiliary function steps ′ plus a constant one. This increment operation at the
end comes from the transition in the abstract machine that evaluates the initial
(Top) context. In steps ′, the number of steps to evaluate an integer is a constant
one, and the number of steps to evaluate an addition is the number of steps to
evaluate each argument plus a constant three. This is more accurate because we
now see the overhead of each addition: if we were counting β-reductions then
this would only have been a single step of evaluation.

3 Extending the Language

We’ve now shown the derivation process for a small test language, but this lan-
guage is not powerful enough to express computations that have interesting time
behaviour. We now extend it with the untyped λ-calculus (variables, abstraction
and application), lists and recursion over lists in the form of fold-right.

These could have been expressed directly in λ-calculus, for example by using
Church encoding instead of integers, but this would introduce an unrealistic
overhead in evaluation. Moreover, a more general recursion operator could have
been introduced instead of fold-right, but for simplicity one tailored to our data
structure, lists, is sufficient. Using fold-right will also simplify the process of
reasoning about time properties, just as it has proved useful for reasoning about
extensional ones [7].



The language is implemented as the following Haskell data type:

data Expr = Var String | Abs String Expr | App Expr Expr
| Add Expr Expr | Val Value
| Cons Expr Expr | Foldr Expr Expr Expr

data Value = Const Int | ConsV Value Value | Nil | Clo String Expr Env

That is, an expression is either a variable, abstraction, application of two ex-
pressions, addition of two expressions, fold-right over an expression (where the
function and empty list-case arguments are both expressions), a list containing
further expressions or a value. In turn, a value is either an integer, list con-
taining further values or a closure (an abstraction paired with an environment
containing bindings for all free variables in the abstraction).

The data type differentiates between expressions and values, in particular
lists that contain unevaluated and evaluated expressions, so that there is no
need to iterate repeatedly over the list to check if each element is fully evaluated,
which would introduce an artificial evaluation overhead.

The primitive functions, such as Add and Foldr , are implemented as fully
saturated, in that they take their arguments all at once. The main reason for this
is to make it easier to define what a value is: if they were introduced as constants
then, for example, App Add 1 would be a value, since it cannot be further
evaluated. This doesn’t affect what can be expressed in the language; partial
application can be expressed by using abstractions, so an equivalent expression
would be Abs "x" (Add 1 (Var "x")), which would be evaluated to a closure.

Note that, for simplicity, the language has not been provided with a type
system. Rather, in this article we assume that only well-formed expressions are
considered.

4 Evaluator

For simplicity, we will consider evaluation using the call-by-value strategy, where
arguments to functions are evaluated before the function is applied. Evaluation
is performed using an environment, that is used to look up what a variable is
bound to. This avoids having to substitute in expressions for variables, which is
complicated by the need to deal with avoiding name-capture. Under call-by-value
evaluation arguments are evaluated before function application, so variables will
be bound to a value. The environment is represented as a list of pairs:

type Env = [(String ,Value)]



The initial evaluator is given below:

eval :: Expr → Env → Value
eval (Val v) env = v
eval (Var x ) env = fromJust (lookup x env)
eval (Abs x e) env = Clo x e env
eval (App f e) env = let Clo x e ′ env ′ = eval f env

v = eval e env
in eval e ′ ((x , v) : env ′)

eval (Add x y) env = let Const m = eval x env
Const n = eval y env

in Const (m + n)
eval (Cons x xs) env = ConsV (eval x env) (eval xs env)
eval (Foldr f v xs) env = case eval xs env of

Nil → eval v env
ConsV z zs → let f ′ = eval f env

x = eval (Foldr (Val f ′) v (Val zs)) env
in eval (App (App (Val f ′) (Val z )) (Val x )) [ ]

That is, values are already evaluated, so they are simply returned. Variables are
evaluated by returning the value the variable is bound to in the environment.
Under the call-by-value strategy, evaluation is not performed under λ-terms, so
abstractions are turned in to values by pairing them with the current environ-
ment to make a closure. An application, App f e, is evaluated by first evaluating
f to an abstraction, then evaluating the body of the abstraction, with the envir-
onment extended with the variable bound to the value that e evaluates to.

Addition is performed by first evaluating both sides to an integer and then
adding them together. This will give another integer result and so does not
require further evaluation. Evaluating a Cons consists of evaluating the first
and second arguments (the head and the tail of the list) and then re-assembling
them using the ConsV constructor to make an evaluated list.

Evaluation of the Foldr case proceeds by first evaluating the list argument
and doing case analysis. If the list is Nil then the result is the evaluation of
the second argument, v . In the non-empty case, first the function argument
is evaluated, then the fold-right applied to the tail of the list, and finally the
function is applied to the head of the list and the result of folding the tail of
the list. This evaluation is performed with an empty environment, since all the
expressions are values at that point.

The evaluation of the fold-right could have been specified in different ways.
The completely call-by-value way would be to evaluate the arguments in left to
right order, so that the first two arguments are evaluated before the list argu-
ment. However, for the Nil list argument case, the function argument to Foldr
is evaluated even though it is not required. The approach in the evaluator is
to evaluate the list argument first to allow pattern matching and then eval-
uate the other arguments depending on what the list evaluated to. So when
the list evaluates to Nil only the second argument to Foldr is evaluated. The
justification not to use the purely call-by-value way is that it would introduce



some artificial behaviour of the Foldr function. When the λ-calculus is extended
with a conditional function, for example, it is not implemented to expand both
branches under call-by-value evaluation, but more efficiently by evaluating the
condition first and then one branch depending on the value of the condition. In
practice, this has little effect because the function supplied to the fold is often
an abstraction and therefore is already evaluated.

5 Complete Function

Performing the derivation process for this extended language (which, as previ-
ously, proceeds by calculation) yields the following steps function:

steps e = steps ′ e [ ] + 1
steps ′ (Val v) env = 1
steps ′ (Var x ) env = 1
steps ′ (Abs x e) env = 1
steps ′ (App f e) env = let (Clo x e ′ env ′) = eval f env

v = eval e env
in steps ′ f env + steps ′ e env +

steps ′ e ′ ((x , v) : env ′) + 3
steps ′ (Add x y) env = steps ′ x env + steps ′ y env + 3
steps ′ (Cons x xs) env = steps ′ x env + steps ′ xs env + 3
steps ′ (Foldr f v xs) env = steps ′ xs env + case eval xs env of

Nil → steps ′ v env + 2
ConsV y ys → let f ′ = Val (eval f env)

x = Val (eval (Foldr (Val f ′) v (Val ys)) env)
in steps ′ f env + steps ′ (Foldr (Val f ′) v (Val ys)) env +

steps ′ (App (App f ′ (Val y)) x ) [ ] + 4

As mentioned earlier, the derived function has calls to the original evaluator,
where the result of evaluation is required. For example, in the Foldr case, a case
analysis has to be performed on the evaluated third argument, to know if it is
empty or not, and so whether to supply the v argument, or to keep folding.

We want to be able to reason about how the time requirements of some ex-
ample functions depends on the size of the arguments to the function. In the case
of functions defined using fold-right, it would be easier to reason about the time
usage if it was expressed as a function over the size of the list, rather than as a
recursive function — the steps function above the Foldr case makes a recursive
call to fold the tail of the list. This can naturally be expressed as a fold-right
over the value list data structure, defined as:

foldrVal :: (Value → b → b)→ b → Value → b
foldrVal f v Nil = v
foldrVal f v (ConsV x xs) = f x (foldrVal f v xs)



Also, if the number of steps to apply the function f to two arguments does
not depend on the value of the arguments, such as adding two expressions, then
a useful further simplification is to express this as a function over the length of
the list argument supplied, defined as:

lengthVal = foldrVal (λ n → n + 1) 0

6 Example Functions

We can now use the derived steps function to look at some examples. Each step
counting function produced was simplified by hand and then QuickCheck [4]
was used to verify that the result produced was equal to the original function,
to check that no errors had been introduced during simplification.

6.1 Summing a List

Summing a list of integers can be expressed using fold-right:

sum [ ] = 0
sum (x : xs) = x + sum xs ⇔ sum xs = foldr (+) 0 xs

The fold-right definition replaces each list constructor (:) with +, and the [ ] at
the end of the list with 0, the unit of addition. First we translate the definition of
sum into the language syntax and then call the steps function on the application
of sum to an argument. The number of steps required to evaluate applying the
sum function to a list of integers, xs, is given below:

steps (App sum (Val xs)) = 21 ∗ (lengthVal xs) + 10

The function can be expressed as a length because the steps required to evaluate
the addition of two values is a constant. The step count is a constant multiplied
by the length of the list argument plus a constant amount; it is directly propor-
tional to the length of the list argument. Though, of more interest is that we
can see the constant factors involved in the evaluation.

6.2 Sum with an Accumulator

An alternative definition of sum is to use an accumulator; the fold-right is used
to generate a function which is applied to the identity function in the empty list
case, and in the non-empty case adds the head of the list to the accumulator.

sumAcc [ ] a = a
sumAcc (x : xs) a = sumAcc xs (a + x ) ⇔ sumAcc xs = foldr f id xs 0

where f x g a = g (a + x )

Using an accumulator could potentially save on space, because additions could
be performed without having to expand the whole list first. It would be useful
to know what effect an accumulator has on the number of steps taken.



Translating the accumulator version and applying steps gives a result of the
same form, linear on the length of the list, but the constant values are larger,
because there is an additional overhead in evaluating the extra abstractions:

steps (App sumAcc (Val xs)) = 26 ∗ (lengthVal xs) + 15

6.3 Concatenation

Concatenating a list of lists can be defined by folding the append function over
the list, and append can also be expressed as a fold-right:

concat xs = foldr append [ ] xs
append xs ys = foldr (:) ys xs

First we need to analyse the append function. The number of steps to evaluate
append applied to two list arguments is as follows:

steps (App (App append (Val xs)) (Val ys)) = 21 ∗ (lengthVal xs) + 15

So the number of steps to evaluate an append is proportional to the length of the
first list argument. The step count of the concat function can now be calculated
using this function. With the step count from the append function inlined, the
resulting steps function is:

steps (App concat (Val xss)) = foldrVal f 10 xss
where f ys s = 21 ∗ (lengthVal ys) + 20 + s

The number of steps required in evaluation is the sum of the steps taken to apply
the append function to each element in the list. If the argument to concat is a
list where all the list elements are of the same length (so the number of steps
taken in applying the append function will always be constant), then this can
be simplified to:

steps (App concat (Val xss)) = 20 + case xss of
Nil → 0
ConsV ys yss → (lengthVal xss) ∗ (21 ∗ (lengthVal ys))

The number of steps is now proportional to the length of the input list multiplied
by the number of steps to evaluate appending an element of the list, which is
proportional to the length of that element.



6.4 Reversing a List

Reversing a list can be expressed directly as a fold-right by appending the re-
versed tail of the list to the head element made in to a singleton list:

reverse [ ] = [ ]
reverse (x : xs) = reverse xs ++ [x ] ⇔ reverse xs = foldr f [ ] xs

where f x xs = xs ++ [x ]

Translating the definition of reverse in to the language syntax and applying the
steps function gives the step count function:

steps (App reverse (Val xs)) = fst (foldrVal g (10,Nil) xs)
where g z (s, zs) = (s + 21 ∗ (lengthVal zs) + 34,

eval (App (App append (Val zs)) (Val (ConsV z Nil))) [ ])

The steps function for reverse is dependent on the steps required to perform the
append function for each element, which is proportional to length of the first
argument to append . The size of this argument is increased by one each time, so
the function is a sum up to the length of the list:

10 +
length xs−1∑

x=0

21x+ 34

Expanding this sum gives the following expression:

10 + 34 ∗ (length xs) +
21 ∗ (length xs − 1) ∗ (length xs)

2
6 c (length xs)2

This is less than a constant multiplied by the square of the length of the list, for
example when c = 11 for all lists of length greater than 47. This means that the
time requirements are quadratic on the length of the list [14].

6.5 Fast Reverse

The reverse function can also be expressed using an accumulator:

fastrev [ ] a = a
fastrev (x : xs) a = fastrev xs (x : a) ⇔ fastrev xs = foldr f id xs 0

where f x g a = g (x : a)

This definition should have better time properties because, as shown above, the
steps required in evaluating the append function is proportional to the length
of the first argument, so appending the tail of the list would be inefficient. The
steps function produced for the accumulator version is directly proportional to
the length of the list:

steps (App fastrev (Val xs)) = 26 ∗ (lengthVal xs) + 15



7 Conclusion and Further Work

We have presented a process that takes an evaluator for a language, and derives a
function that gives an accurate count of the number of steps required to evaluate
expressions using an abstract machine for the language. Moreover, all the steps
in the derivation process are purely calculational, in that the required function
at each stage is calculated directly from a specification of its desired behaviour.

Using an extended λ-calculus under call-by-value evaluation, the examples
in the previous section give the expected complexity results, but also show the
constants involved. This is useful to know because of the additional overheads
that functions of the same complexity may have, for example in summing a list
with and without an accumulator. They also show the boundaries at which one
function with a lower growth rate but larger constants becomes quicker than
another, for example in the two different definitions of the reverse function.

Ultimately, the most accurate information is to do real timing of programs.
The approach taken in the profiler [12] that is distributed with the Glasgow
Haskell Compiler, is similar to that here, in that abstract machines are used to
bridge the gap between a big-step cost semantics and a small-step implement-
ation. A regular clock interrupt is used to collect time information and assign
to the costs of functions. However, this is a profiler, and we want to be able to
express costs as a function over the arguments supplied, and not just in terms
of hard time measurements. The thesis of this paper is that we can obtain this
useful information relatively easily.

This work could be extended by considering more complicated evaluation
strategies, such as call-by-name or lazy evaluation. It would also be interesting
to apply the same technique to look at the space requirements for functions, by
measuring the size of the additional data structures produced at the abstract
machine level. Finally, a useful addition to this work would be to develop a
calculus to automate deriving the step functions.

